#!/usr/bin/env perl
######################################################################
#
# Copyright 2002-2020 by Wilson Snyder. This program is free software; you
# can redistribute it and/or modify it under the terms of either the GNU
# Lesser General Public License Version 3 or the Perl Artistic License
# Version 2.0.
# SPDX-License-Identifier: LGPL-3.0-only OR Artistic-2.0
#
######################################################################
# DESCRIPTION: Edits flex output to get around various broken flex issues.

use warnings;

my $Opt_Prefix = $ARGV[0] or die "%Error: No prefix specified,";

foreach my $line (<STDIN>) {
    # Fix flex 2.6.1 warning
    $line =~ s/for \( i = 0; i < _yybytes_len; \+\+i \)/for ( i = 0; (yy_size_t)(i) < (yy_size_t)(_yybytes_len); ++i )/g;
    # Fix flex 2.6.0 warning
    $line =~ s/\(\(int\) \(\(yy_n_chars\) \+ number_to_move\) > YY_CURRENT_BUFFER_LVALUE->yy_buf_size\)/((int) ((yy_n_chars) + number_to_move) > (int) YY_CURRENT_BUFFER_LVALUE->yy_buf_size)/g;
    $line =~ s/ number_to_move == YY_MORE_ADJ / (int)number_to_move == (int)YY_MORE_ADJ /;
    # Fix flex 2.5.4 namespace omission
    $line =~ s/^class istream;/\#include <iostream>\nusing namespace std;\n/;
    # Fix flex 2.5.31 redefinition
    $line =~ s!(\#define\s+yyFlexLexer\s+yyFlexLexer)!//flexfix: $1!g;
    # Fix flex 2.5.1 yytext_ptr undef
    $line =~ s!(\#undef\s+yytext_ptr)!//flexfix: $1!g;
    # Fix flex 2.5.4 and GCC 4.1.0 warn_unused_result
    $line =~ s!\(void\) *fwrite\((.*)\)!if (fwrite($1)) {}!g;
    # Fix flex 2.5.33 and GCC 4.1.2 "warning: comparison between signed and unsigned integer expressions" in YY_INPUT
    $line =~ s!for \( n = 0; n < max_size && !for ( n = 0; ((size_t)n < (size_t)max_size) && !g;
    # Fix flex 2.5.4 and GCC 4.0.2 under FLEX_DEBUG
    $line =~ s!--accepting rule at line %d !--accepting rule at line %ld !g;
    # Fix compiler warning filenames
    $line =~ s!(#line \d+ ".*)_pretmp!$1!;

    print "$line";
}
